

UNIVERSITY OF ZAGREB
FACULTY OF ELECTRICAL ENGINEERING AND
COMPUTING

MASTER THESIS No. 2568

Design of a strongly-typed programming language

Petar Mihalj

Zagreb, May 2021.

MASTER THESIS ASSIGNMENT No. 2568

Student: **Petar Mihalj (0036497900)**

Study: Computing

Profile: Computer Science

Mentor: asst. prof. Ante Đerek

Title: **Design of a strongly-typed programming language**

Description:

Programming languages play a central role in the development of software. However, there is no silver bullet programming language, which would be efficient, easy to use, portable and consistent. The goal of this graduate thesis is to design a new programming language with the emphasis on how strong typing can address common drawbacks. The language specification should be described in a formal form. Design should be followed-up by a careful analysis of some aspects of the language, for example to prove the desirable properties of the type system. The final goal of the thesis is to implement the tool chain, which primarily consists of a compiler, but can include other tools, such as language servers and formatters. The thesis should be accompanied with the source code of all developed software. All references should be clearly cited. Any assistance received should be clearly acknowledged.

Submission date: 28 June 2021

DIPLOMSKI ZADATAK br. 2568

Pristupnik: **Petar Mihalj (0036497900)**

Studij: Računarstvo

Profil: Računarska znanost

Mentor: doc. dr. sc. Ante Đerek

Zadatak: **Oblikovanje strogo tipiziranog programskog jezika**

Opis zadatka:

Programski jezici centralna su karika u razvoju programske potpore. Ipak, ne postoji programski jezik najbolji za sve primjene, u isto vrijeme efikasan, jednostavan, prenosiv i konzistentan. Cilj ovog diplomskog rada dizajnirati je novi programski jezik, s naglaskom na prednosti strogih tipova u rješavanju čestih problema. Specifikacija jezika treba biti opisana u formalnom obliku. Nakon dizajna jezika potrebno je provesti pažljivu analizu nekih svojstava jezika, na primjer pokazati poželjna svojstva sustava tipova. Posljednji cilj rada implementacija je skupa alata, kojeg primarno čini prevoditelj, ali koji može uključivati i druge alate, poput jezičnih poslužitelja (language servers) i formatera koda. Rad treba uključivati izvorni kod razvijenih alata. Sve reference trebaju biti jasno citirane. Sva primljena pomoć treba biti jasno obznanjena.

Rok za predaju rada: 28. lipnja 2021.

I thank everybody...

CONTENTS

1. Introduction	1
1.1. Example program	2
1.2. Paper organization	4
2. The AGT Programming Language	5
2.1. Lexical analysis	5
2.1.1. Literals	7
2.1.2. Identifiers	8
2.1.3. Operators	8
2.1.4. Special operators	9
2.2. Syntactical analysis	10
2.3. Semantic analysis	12
2.4. Type inference	13
2.5. Concrete type request	14
2.6. Function type request	14
2.7. Lifetime semantics	16
3. AGT program examples	20
3.1. Limited parametric polymorphism	20
3.2. Compile time computation	21
3.3. Ownership semantics	22
3.4. Memory primitives	25
3.5. Memory allocation	25
3.6. Object oriented example	26
4. Future improvements	30
4.1. References	30
4.2. Syntactic improvements	30

4.3. Multiple source file support	30
4.4. Portability	30
5. Conclusion	31
Bibliography	32

1. Introduction

The programming language developed as a part of this thesis is called AGT. The name AGT stands for an unfortunate fact that the names of all precious stones are taken by other programming languages, hence "All Gems are Taken".

The AGT programming language is a statically and strongly typed language, with a highly expressive type system. The type system is used for three main purposes:

1. types determine the memory footprint of values
2. types allow for polymorphic behaviour of operators and functions on all *concrete types* (built-in or struct types)
3. type system lets the programmer perform compile-time computation

The language also allows for implementation of object lifetime constructs, such as those seen in languages as C++ or Rust; object creation, copying and destruction.

The reference AGT compiler (AGTC) produces executable binaries only. This is in sharp contrast to some other languages, which can produce object files, to later be linked into executables for a particular runtime, or used to augment the runtime environment itself (kernel modules, for example). This decision was made primarily because the goal of the thesis was to study the language from the application programmer's perspective. This goal implies that AGT is running on a rich runtime environment, where heap memory management and standard I/O are available. Future updates to AGT specification are meant to allow AGT to be compiled into linkable files compatible with various platform specific ABIs.

The compiler frontend is implemented using Python3 programming language, while the backend uses LLVM compiler infrastructure (Lattner und Adve, 2004). Allowing for some simplification, AGT is first compiled into LLVM intermediate representation (LLVM-IR), and then is turned into executable by a chain of tools for compiling LLVM-IR and linking the resulting object files into executables.

1.1. Example program

Let us first check out an example AGT program in listing 1.1:

Listing 1.1: Example AGT program

```
1  fn outnl(i){
2      out(i);
3      out('\n');
4  }
5
6  fn mul(a, b) -> a{
7      return a*b;
8  }
9
10 fn main()->i32{
11     let a = in<i32>();
12     let b = in<i8>();
13
14     let c = mul(a, cast<i32>(b));
15     outnl(a);
16     outnl(b);
17     outnl(c);
18 }
```

Every AGT program has to have a function definition named `main`, which returns a 32-bit-wide integer, and has no parameters. Some functions, like `outnl`, don't return anything. It is important to note that you can't specify that a function returns `void` like in C for example.

A `let <x> = <y>;` construct is an *initialization assignment statement*. This roughly translates to an allocation of memory on the stack (which can be referred to as the *location* of `<x>` from now on), and copying the value of expression `<y>` to the location of identifier `<x>`.

Note that the function call `in` (line 11) is *parametrized* by a *type argument* `i32`. Type arguments are used to pass types, but not values. They are different than *value arguments*, which carry value too, along with a type. Note that *passing* of type parameters is purely a compile-time concept; types can't be referred to during runtime.

The function `in` is a builtin, and the `i32` is used to signal the compiler that we want an instance of this function which returns an `i32`, instead of a `bool`, for example.

Of course, user-defined functions can also have *type parameters*, along with *value parameters*; we will get to these later.

Notice that we used the `cast<i32>` builtin function to convert `b` to `i32` before passing it to function `mul`. If we didn't do that, the compiler would signal that it can't compute the expression `a*b`, since multiplication is predefined only for integer types of same size. Apart from the `i8` and `i32` we used in this example, `i16` and `i64` are also available.

Also, line 6 is terminated by a `-> a`. This is a return type specification; it states that this function will return whatever is type of `a`.

You might have noticed a certain feature while inspecting the code; namely the absence of explicitly stated types in the signature of function definition `mul`. While some *dynamically typed* languages (ex. Python) allow for object of any type to be passed to the function, AGT behaves quite differently.

Every time AGTC (AGT compiler) encounters a function call, it tries to infer a *function type*. Function type is inferred from function definitions. Consequently, you can think of the definitions in source code more as *templates* for synthesis of actual code, than representation of code. These definitions lack types, and can't be considered in isolation. In this example, the call of function `mul` causes the compiler to try to infer a function type.

Parameters of this `mul` function definition (`a`, `b`) can be fit with objects of any type; it is after this fitting that the compiler will determine that passed types can or can't be used (due to their incompatibility with a function definition body, for example).

It is important to emphasize that one function definition can be a source of many function types. The `outnl` (output with newline) function is a prime example of this behaviour. It is called three times, with argument of types `i32`, `i8` and `i32`. Thus, the compiler has to infer two function types, one which can be fed with an `i8`, and one with `i32`. The success of this inference corresponds with compilers ability to infer functionality of body of the given function definition.

In this case, inference of function definition body will be possible if both function calls on lines 2 and 3 can be inferred. Since `out` function is a built-in for both `i32` and `i8`, line 2 won't be problematic for inference. Line 3 will be inferred correctly if AGTC succeeds to resolve function `out` which takes in an argument of type `char`. Since this is also a built-in, inference of function types for both version of `mul` succeeds.

The property of AGT **function calls** which can make them refer to different function types, when called with different sets of arguments, is called *polymorphism*. Some sources label polymorphic the very **function definitions** which, when called upon,

can result in such behaviour. We will use these conventions interchangeably, since the definition itself is non-rigorous.

Function definition of `outnl` introduces a special type of polymorphism, *parametric polymorphism*. This polymorphic behaviour is said to be present if the function definition can be a source for potentially unlimited number of types, as long as those types are compatible with function definition body.

Python programming language exhibits a superficially similar behaviour, but its function body compatibility checks occur at runtime. The programming style which utilizes this language capability is often called "Duck Typing" (Fred L. Drake Jr.).

1.2. Paper organization

This master thesis will gradually introduce the reader to AGT.

Chapter 1 has already introduced the reader to basic syntax and semantics, using a simple example and emphasizing the surface-level properties of type system and inference.

After the surface-level exploration of the language, chapter 2 will provide an in-depth description of AGT. This chapter will have many references to the AGTC reference implementation, rather than to a separate formal specification. Since AGT is still not a finalized product, formal specification does not exist, and its functionality is determined by a reference compiler. We will mostly focus on the type system, since it is unorthodox and is the main contribution of the author. Author will justify the decisions which had been made on all levels of AGT design process.

Chapter 3 will supplement the reader with plenty of examples of AGT code. These will further deepen reader's understanding of AGT. The examples describe implementation of common programming paradigms, constructs, patterns in AGT, with a detailed explanation of more esoteric parts of the source code.

In chapter 4 we will address the issues AGT has both on definition and implementation levels, and discuss the various features that can be improved or added to the language.

Chapter 5 will conclude the thesis and describe future work the author will undertake in order to improve the language.

2. The AGT Programming Language

AGT compilation process consists of 4 main phases:

1. Lexical analysis - tokenization of input source code text
2. Syntactical analysis - rule-based grouping of tokens into an *abstract syntax tree*
3. Semantical analysis - transforming the syntax tree into a *semantical tree*
4. Code generation - inference of the `main` and code generation

The process is successful if all the steps succeed.

In this chapter we will describe each phase in great detail, contrasting the newly defined terms against their use in other languages.

2.1. Lexical analysis

The AGT Lexical analysis phase is a standard regular expression parsing process. Lexical analyzer is fed a stream of characters, which get grouped into tokens, according to regular expressions and ambiguity resolution rules.

Lexical analyzer tool used for the AGTC is PLY - Python Lex-Yacc (Beazley, a). Python Lex-Yacc is a tool which tries to bring the functionality of well known lexical and syntactical analyzers Lex and Yacc into Python. The main difference is that PLY does not *generate* the concrete lexer and parser. Instead, PLY user defines the rules fed to PLY by specifying programming constructs like functions that correspond to these rules. PLY then inspects these constructs using reflective capabilities of Python, and acts according to these rules when performing lexical or syntactical analysis.

Let us check out the main lexing module of AGTC, which specifies rules, states and precedence that PLY uses to perform lexical analysis for AGT. Some parts of the lexing module are left out, which is indicated with `# . . .` in source code.

Listing 2.1: heyy

```
1  class Lexer():
2      states = (
3          ('mlc', 'exclusive'), ('strchar', 'exclusive'),
4      )
5      #...
6      tokens = (
7          'INTL', 'BOOLL', 'ID',
8          'IF', 'ELSE', 'BREAK', 'FOR', 'WHILE', 'RETURN',
9          #...
10     )
11     t_ADD = r'\+'
12     t_SUB = r'\-'
13     t_MUL = r'\*'
14     #...
15     reserved = {
16         'type': 'TYPE',
17         'fn': 'FN',
18         'let': 'LET',
19         #...
20     }
21     def t_ID(self, t):
22         r'[a-zA-Z_][a-zA-Z_0-9]*'
23         t.type = self.reserved.get(t.value, 'ID')
24         return t
25
26     def t_INTL(self, t):
27         r'(\d+)(li8|li16|li32|li64)'
28         return t
29     #...
```

This lexer module has 3 states: `main` (implicit), `mlc` (for multiline comments) and `strchar` (for *char arrays* or char literals).

Rules for simple operators are given in a string form, in a form of `t_<TOKEN> = <REGEX>`.

Rules for more tokens which are defined with more complex regular expressions, like int literals or identifiers, are defined using functions. Those functions can also alter the behaviour of the lexer when such a token is parsed. For example, whenever

an identifier is parsed, the function `t_ID` looks up whether the identifier is in a list of reserved words, and if it is, changes the token's type. This neat trick greatly simplifies regular expressions, and is taken from official PLY documentation (Beazley, b).

PLY documentation states that priority is given to regular expressions defined by functions, in the order in which they are given. After function-based rule definitions, priority is given to string defined rules (the ones with greater regex length first).

We employ the *state* feature of PLY lexing module, which allows us to change the lexer's behaviour when it encounters, for example, a beginning of a multiline comment. In that particular case, lexer goes into `mlc` state. When in `mlc` state, lexer parses and caches every symbol until it gets to the end of the comment. Both type of comments, `/*multiline comment*/` and `//singleline comment` are discarded after being parsed.

2.1.1. Literals

String and character literals are given in double and single quotes, respectively. They can include escaped characters: (`\0`, `\n`, `\t`, `\"`, `\'`). Character literal must consist of exactly one character or escaped character.

Integer literals are characterized by both the value and the size of integer containing this value. Possible integer sizes are 8, 16, 32 and 64.

Example integer literals are:

- 5 (implicitly i32)
- 5i32
- 516
- 5i8
- 5i64
- 324243 (implicitly i32)

Boolean literals are:

- true
- True
- false
- False

2.1.2. Identifiers

Identifiers consist of a letter or an underscore, followed by any number of letters, underscores or digits. Reserved words are excluded from identifiers.

Reserved words are:

- 'fn', 'struct',
- 'let', 'type',
- 'if', 'else', 'break', 'for', 'while', 'return',

2.1.3. Operators

We will now list the operator families, without discussing their semantics in detail, because

- Built-in versions of these operators will be listed in later chapters, and the reader can intuitively reason about which ones are available.
- User-defined versions of these operators can adhere to a number of different semantics (which is not recommended, but is possible).
- Operators can be applied on types during compile time (to produce other types). This discussing is better to be left to later chapters.

Apart from the operators themselves, we will list their special names. These names will be used to translate operator applications to function calls. For example, when `+` is applied to value 3 and 5 (`3+5`), compiler will insert a call to `__add__(3, 5)` instead. This method derived from operator's special name is an example of what we call a *dunder* or *magic* method (according to Python folklore). *Dunder* comes from double-underscore which both precedes and follows the special name.

Arithmetic operators	Comparison operators	Boolean operators
<code>+</code> (add)	<code>==</code> (eq)	<code>&</code> (and)
<code>-</code> (sub)	<code>!=</code> (ne)	<code> </code> (or)
<code>*</code> (mul)	<code>>!</code> (gt)	<code>~</code> (not)
<code>/</code> (div)	<code><!</code> (lt)	
<code>%</code> (mod)	<code>>=</code> (ge)	
	<code><=</code> (le)	

Note that, even though we call the last group the *boolean* operators, those operators can be evaluated on operands which are not booleans, as long as the corresponding

dunder functions are defined in the program.

The same goes for arithmetic and comparison operators; a programmer can use builtins functions for all integer types, but can define additional dunder functions for `rectangle` objects, for example.

2.1.4. Special operators

Special operators can't be user-defined, and have a fixed meaning:

- *Address of operator* (`@`) can be used on any L-value (explained in later chapters), and evaluates to a pointer to that value. For example, if `i` is an `i8`, then `@i` is a pointer to `i8`. We can also denote this pointer as `@i8` - exactly the way it can be used in the context of a type. For example, if you want to check whether a function argument is of `@i8` type (to get polymorphic behaviour), you would start this function with a line: `type _ = enable_if<arg == @i8>;`.
- *Dereference operator* (`!`) can be used on any pointer value, and evaluates to a value which is being pointed to. For example, if `p` is an `@i8`, then `p!` is a value of type `i8`. This operator can also be used in the context of a type. For example, if you want to check whether a function argument is a pointer (to whatever type), and capture the pointed-to type, you would start this function with a line: `type pointed_to = enable_if_resolve<arg!>;`. A new special type expression `enable_if_resolve<...>` exhibits similar behaviour as `enable_if<...>`, in sense that it lets you create polymorphic behaviour. This one will succeed if the argument is successfully evaluated (it doesn't have to evaluate to `i1`). If it succeeds, it will also act as if it wasn't there. In this case,
- *Index (bracket) operator* must be used on a pointer, and be given any integer type as an argument. It will offset the pointer depending on the underlying type. For example, if `p` is a `char` pointer, `p[5]` will evaluate to a pointer with a location $lp + 5 \cdot s$, where lp is `p`'s value and s is size of `char` type.

Precedence rules of operators will be stated in section 2.2. An interested reader will find more details concerning the implementation in the source code supplied with this paper.

2.2. Syntactical analysis

Syntactical analysis is achieved with a LALR parser implemented in PLY package.

Tokens resulting from lexical analysis are fed into a stack machine which, depending on the state of the top of the stack and next token in the list, either pushes the next token on top of the stack (*shift*) or converts a number of tokens on the top of the stack to another token (*reduce*).

The grammar rule specifications are given in a different way than the one standard for PLY. PLY uses functions by default, while AGTC uses classes which have to adhere to certain structure. These class definitions are turned into functions (using reflective programming) and finally dynamically inserted into syntactical parser class definition. A reader who is interested in details can consult the source code.

The output of a syntactical analysis phase is an abstract syntax tree - a tree structure which denotes applications of grammar rules to tokens. An example syntax tree, which results from the statement `let n = in<i32>();` is given:

Listing 2.2: a

```
1 Statement(  
2 - statement = InitStatement(  
3 - - name = Expression(  
4 - - - expr = IdExpression(  
5 - - - - id = str(n)  
6 - - - )  
7 - - )  
8 - - expr = Expression(  
9 - - - expr = UnaryExpression(  
10 - - - - expr = ParenthesesCallExpression(  
11 - - - - - expr = Expression(  
12 - - - - - - expr = UnaryExpression(  
13 - - - - - - - expr = AngleCallExpression(  
14 - - - - - - - - expr = Expression(  
15 - - - - - - - - - expr = IdExpression(  
16 - - - - - - - - - - id = str(in)  
17 - - - - - - - - - )  
18 - - - - - - - - )  
19 - - - - - - - - - expr_list = [  
20 - - - - - - - - - Expression(  
21 - - - - - - - - - - - expr = IdExpression(  

```



```

22 - - - - - id = str(i32)
23 - - - - - )
24 - - - - - )
25 - - - - - ]
26 - - - - - )
27 - - - - - )
28 - - - - - )
29 - - - - - expr_list = [
30 - - - - - ]
31 - - - - - )
32 - - - )
33 - - )
34 - )
35 )

```

This tree consists of a single statement. We won't get into details of the semantics here, since there is a another step which will contextualize the syntax tree and resolve some ambiguities, which couldn't have been resolved with a LALR parser (which PLY uses).

An example of a rule which is used to generate binary expression nodes is:

Listing 2.3: a

```

1 class BinaryExpression(ParserRule):
2     """ BinaryExpression : Expression ADD Expression
3                             | Expression SUB Expression
4                             | Expression MUL Expression
5                             | Expression DIV Expression
6                             | Expression MOD Expression
7                             | Expression LE Expression
8                             | Expression GE Expression
9                             | Expression LT Expression
10                            | Expression GT Expression
11                            | Expression EQ Expression
12                            | Expression NE Expression
13                            | Expression AND Expression
14                            | Expression OR Expression
15     """
16
17     def __init__(self, r):

```

```

18         self.left = r[0]
19         self.op = r[1]
20         self.right = r[2]

```

The shift/reduce conflicts of arithmetic expressions are resolved by specifying a list of operator priorities.

2.3. Semantic analysis

The need for semantical analysis is caused by a fact that AGT's syntax is highly contextual, that is, expressions that appear in one place can have a radically different meaning than same expressions which appear in another place.

For example, in the following code snippet, the expression `a` is used in two distinct contexts.

Listing 2.4: `a`

```

1 Statement(
2 - statement = InitStatement(
3 - - name = Expression(
4 - - - expr = IdExpression(
5 - - - - id = str(n)
6 - - - )
7 - - )
8 - - expr = Expression(
9 - - - expr = UnaryExpression(
10 - - - - expr = ParenthesesCallExpression(
11 - - - - - expr = Expression(
12 - - - - - - expr = UnaryExpression(
13 - - - - - - - expr = AngleCallExpression(
14 - - - - - - - - expr = Expression(
15 - - - - - - - - - expr = IdExpression(
16 - - - - - - - - - - id = str(in)
17 - - - - - - - - - )
18 - - - - - - - - )
19 - - - - - - - - expr_list = [
20 - - - - - - - - - Expression(
21 - - - - - - - - - - expr = IdExpression(
22 - - - - - - - - - - - id = str(i32)

```

```

23 - - - - - - - - - - )
24 - - - - - - - - - - )
25 - - - - - - - - - - ]
26 - - - - - - - - - - )
27 - - - - - - - - - - )
28 - - - - - - - - - - )
29 - - - - - - - - - - expr_list = [
30 - - - - - - - - - - ]
31 - - - - - - - - - - )
32 - - - - - - - - - - )
33 - - - - - - - - - - )
34 - - - - - - - - - - )
35 - - - - - - - - - - )

```

In the ending of the first line, `a` is a type expression, that is, a compiler is required to produce a type (which the function returns). In the second line, `a` is a value expression, which means compiler is required to produce a value of that expression, rather than a type. We haven't yet specified what means to produce either a type or a value, but those implementation details will be dealt with in the next chapter.

What is important for the understanding of AGT is that

Listing 2.5: a

```

1 @ dataclass
2 class InitStatement( FunctionStatement ):
3     name: str
4     expr: ValueExpression
5
6 @ dataclass
7 class MemberDeclarationStatement( StructStatement ):
8     name: str
9     type_expr: TypeExpression

```

2.4. Type inference

Type inference is the most complex phase of AGT compilation process. Before we go heads first into it's description, we will lay down some definitions and motivate them. Type inference also includes LLVM-IR generation.

AGT's type inference engine works as a functional programming language, in a sense that it has to evaluate *type requests*. This evaluation is *pure*; repeated evaluations always yield the same result. In a sense, the type engine is a box which provides two evaluation functions.

2.5. Concrete type request

The first function is a *Concrete type request* (CTR). This function maps a name and a list of concrete types (*type arguments*) into a new concrete type. For example,

1. CTR("char", []) -> CharType
2. CTR("i32", []) -> IntType(32)
3. CTR("rectangle", [i32]) -> RectangleType(with i32 side lengths)

How does the type engine achieve this mapping?

One way is to use `concrete type generators`. They take CTR arguments and decide whether they can provide a type, for example, a bool type generator can check whether the name is equal to "bool" and whether there are no supplied type arguments; if both conditions are satisfied, it can provide the concrete type.

The other way is to iterate over user-supplied struct definitions. For every struct definition whose name matches the supplied name, and whose number of type parameters matches the number of supplied type arguments, the inference is being done. Every successful inference is considered in a final decision.

Inference using struct definitions consists of evaluating the body of struct definition, which might need recursive calls to type engine itself. We ensure that these recursions are taken care of; for example, recursive structure definitions will fail.

When both of these methods finish, there has to be **exactly one** candidate type, otherwise, we have an ambiguity, which is an error.

2.6. Function type request

The second inference function is a *Function type request* (FTR). This function maps a name and **two** lists of concrete types (*type arguments* and *value arguments*) into a new concrete type. For example,

1. FTR("main", [], []) -> FunctionType

2. FTR("fibonacci", [], [i32]) -> FunctionType
3. FTR("power", [fast], [i32, i32]) -> FunctionType
4. FTR("power", [slow], [i32, i32]) -> FunctionType

A FunctionType is a rather complex type; it needs to have all information needed for code synthesis, including LLVM code and return type.

By default, every function call is preceded by copying of its arguments. However, some built-in functions, like copy functions of pointers, should not copy their arguments (because of infinite recursive loops). Only builtin functions can ignore this copying behaviour, and the flag which indicates whether to do so is also a part of a FunctionType.

Let us take a look at the following code snippet, which demonstrates a sort of polymorphic behaviour.

Listing 2.6: a

```

1  struct slow {};
2  struct fast {};
3
4  fn power<T>(b, p){
5      type _ = enable_if<T==slow>;
6      ...
7  }
8
9  fn power<T>(b, p){
10     type _ = enable_if<T==fast>;
11     ...
12 }
```

Suppose the type engine is invoked to fulfill a FTR("power", [slow], [i32, i32]). Since "power" is not in builtin functions, TE will resort to using function definitions.

When TE starts to infer a type from the definition at line 4, it will immediately encounter a special construct `enable_if<...>` at line 5. This construct is the main driver of polymorphism in AGT. AGTC will first evaluate its argument, `T==slow`, which will resolve to `i1` (equivalent of true when used in type context). `Enable_if` will, since its argument represents truth, act as if didn't exist; it will just assign the type to variable on the left side. Since the variable is a *discard token* (`_`), the type won't be assigned, but rather discarded.

The user has achieved polymorphic selection using `enable_if`. He signalled the compiler that this definition of `power` function fits programmers intentions.

On the other hand, consider the definition of `power` at line 9. In this line, `enable_if` argument will evaluate to `!0` (false in type context). This will cause the TE to abandon this function as a potential candidate for type resolution. Note that this behaviour is not a error (the one which AGTC reports as a mistake), but rather a mechanism for polymorphic selection.

2.7. Lifetime semantics

Lifetime semantics is a broad term which roughly encompasses the following aspects of objects in programming languages:

1. Objects exist in a specific context, which is dependant on their
lexical environment (the scope)
runtime (the function invocation)

For example, parameters of the function exist and can be referenced throughout the function body, but stop existing when that specific invocation of the function terminates.

2. The context in which objects exist is called their *lifetime*
3. Object's lifetime starts with *initialization* (*construction* in some sources).

Having precise control over construction can allow programmer to restrict usage of the object or set up invariants

4. Object's lifetime ends with *destruction*. Having precise control over destruction allows programmer to deconstruct complex inner workings of the object, without relying on programmer to do it manually. For example, objects can *release* their resources when they stop existing.
5. Transferring or duplicating object's contents to other memory location can also be controlled, via `copy` operation. This allows us to, for example, deep copy the memory array, instead of shallow copying the pointer and thus incidentally creating unwanted memory aliases.

Various other semantics can be implemented on top of lifetime model, we will check out an example called `Ownership semantics` in the last section. Now we

will describe how lifetime semantics fit into definition of AGT. Before we get into the details though, we have to thoroughly define types of values.

Definition 1 (L-value). A value expression is said to a L-value, if the value it evaluates to **can** be referred to from other context.

Definition 2 (R-value). A value expression is said to a R-value, if it is not L-value, that is if the value it evaluates to **can't** be referred to from other context.

For example, every statement `let <v> = . . .` creates a new memory location, so the expressions `<v>` following it and referring to same identifier `<v>` are lvalues.

On the other hand, function call expressions such as `gcd(3, 2)` are rvalues; their result (function return value) can't be referred to from any other place, apart from this one.

The distinction between L-values and R-values is crucial when reasoning about copy operations. Since R-values can't be referred to from any other context, instead of copying their value, we can *memory copy* their value. The crucial distinction is that *copying* involves calling the `copy` function (which might be very expensive, for example for vectors), and *memory copying* does not copies only the literal memory contents.

Definition 3 (Initializer). An initializer is a function whose name is *init, and which has at least one value parameter; a pointer*

Definition 4 (Copy function). An copy function is a function whose name is *copy, and which has two value parameters;*

Definition 5 (Destructor). An destructor is a function whose name is *dest, and which has only one value parameter; a pointer*

Check out the following program and it's output:

Listing 2.7: a

```
1 fn outnl(i){
2     out(i);
3     out('\n');
4 }
5
6 struct B{
7     let member = i32;
8 }
9
10 fn __init__(bp, i){
11     type _ = enable_if<bp==@B>;
```

```

12         type _ = enable_if<i==i32>;
13
14         outnl("Init B started");
15         bp!.member = i;
16         outnl("Init B finished");
17     }
18
19     fn __copy__(bp1, bp2){
20         type _ = enable_if<bp1==@B>;
21         type _ = enable_if<bp2==@B>;
22
23         outnl("Copy B started");
24         bp1!.member = bp2!.member;
25         outnl("Copy B finished");
26     }
27
28     fn __dest__(bp){
29         type _ = enable_if<bp==@B>;
30
31         outnl("Dest B started");
32         __dest__(@bp!.member);
33         outnl("Dest B finished");
34     }
35
36     fn main() -> i32{
37         let b = object<B>(5);
38         outnl(b.member);
39
40         return 0;
41     }

```

Listing 2.8: a

```

1 Init B started
2 Init B finished
3 5
4 Dest B started
5 Dest B finished

```

You can clearly see that we have used `enable_if` construct in lifetime functions

too. This is crucial to prevent multiple candidates for different concrete types.

These were the basics of lifetime semantics, we will get to more details in examples section.

3. AGT program examples

This section of the paper will introduce the reader to AGT by presenting, explaining and contemplating about example programs.

3.1. Limited parametric polymorphism

Parametric polymorphism is a possibility of function to be called with different sets of argument types.

Check out the following example:

Listing 3.1: a

```
1  fn outnl(i){
2      out(i);
3      out('\n');
4  }
5
6  struct is_int<T> -> R{
7      type R = (T==i64 | T==i32 | T==i16 | T==i8);
8  }
9
10 fn fib<T>(n) -> T{
11     type _ = enable_if<is_int<T>>;
12     type _ = enable_if<n == i32>;
13
14     if (n<=2){
15         return cast<T>(1);
16     }
17     else{
18         return fib<T>(n-1)+fib<T>(n-2);
19     }
```

```

20 }
21
22 fn main() -> i32 {
23     let t = 12;
24     let f1 = fib<i8>(t);
25     let f2 = fib<i32>(t);
26     outnl(f1);
27     outnl(f2);
28     return 0;
29 }

```

Output:

Listing 3.2: a

```

1 -112
2 144

```

3.2. Compile time computation

Check out the following example:

Listing 3.3: a

```

1 struct fib<T> -> i1 {
2     type _ = enable_if<T == i1>;
3 }
4 struct fib<T> -> i1 {
5     type _ = enable_if<T == i2>;
6 }
7
8 struct fib<T> -> R {
9     type _ = enable_if<T != i1>;
10    type _ = enable_if<T != i2>;
11    type R = fib<T-i1>+fib<T-i2>;
12 }
13
14 fn main() -> i32 {
15     let v = type_to_value<fib<i12>, i32>();
16     out(v);
17     return 0;

```

```
18 }
```

Output:

Listing 3.4: a

```
1 144
```

3.3. Ownership semantics

Check out the following example:

Listing 3.5: a

```
1 fn outnl(a){
2     out(a);
3     out("\n");
4 }
5
6 struct shared_ptr<T>{
7     let item_ptr = @T;
8     let count = @i32;
9 }
10
11 fn __init__(sptr_ptr , item_ptr){
12     outnl("INIT");
13     type T = enable_if_resolve<sptr_ptr!.T>;
14     type _ = enable_if<sptr_ptr == @shared_ptr<T>>;
15     type _ = enable_if<item_ptr == @T>;
16
17     sptr_ptr!.item_ptr = item_ptr;
18     sptr_ptr!.count = heap_object<i32>(1);
19 }
20
21 fn __dest__(sptr_ptr){
22     outnl("DEST");
23     type T = enable_if_resolve<sptr_ptr!.T>;
24     type _ = enable_if<sptr_ptr == @shared_ptr<T>>;
25
26     sptr_ptr!.count! = sptr_ptr!.count! - 1;
27     if (sptr_ptr!.count! == 0){
```

```

28         outnl("RELEASE");
29         heap_free(sp_ptr_ptr!.item_ptr);
30         heap_free(sp_ptr_ptr!.count);
31     }
32 }
33
34 fn __copy__(sp_ptr_ptr_dest, sp_ptr_ptr_src){
35     outnl("COPY");
36     type T = enable_if_resolve<sp_ptr_ptr_dest!.T>;
37     type _ = enable_if<sp_ptr_ptr_dest == @shared_ptr<T>>;
38     type _ = enable_if<sp_ptr_ptr_src == @shared_ptr<T>>;
39
40     sp_ptr_ptr_src!.count! = sp_ptr_ptr_src!.count! + 1;
41
42     sp_ptr_ptr_dest!.item_ptr = sp_ptr_ptr_src!.item_ptr;
43     sp_ptr_ptr_dest!.count = sp_ptr_ptr_src!.count;
44 }
45
46 fn make_shared(item_ptr) -> shared_ptr<inner>{
47     type inner = enable_if_resolve<item_ptr!>;
48     return object<shared_ptr<inner>>(item_ptr);
49 }
50
51 // actual use
52
53 fn main() -> i32{
54     let a = make_shared(heap_object<i8>(15i8));
55     outnl("Before B is initialized");
56     outnl(a.count!);
57     outnl(a.item_ptr!);
58     {
59         let b = a;
60         outnl("After B is initialized");
61         outnl(a.count!);
62         outnl(a.item_ptr!);
63         outnl(b.count!);
64         outnl(b.item_ptr!);
65         b.item_ptr! = 13i8;

```

```

66         outnl(" After B is changed ");
67         outnl(a.count!);
68         outnl(a.item_ptr!);
69         outnl(b.count!);
70         outnl(b.item_ptr!);
71     }
72     outnl(" After B is destructed ");
73     outnl(a.count!);
74     outnl(a.item_ptr!);
75
76     return 0;
77 }

```

Output:

Listing 3.6: a

```

1  INIT
2  Before B is initialized
3  1
4  15
5  COPY
6  After B is initialized
7  2
8  15
9  2
10 15
11 After B is changed
12 2
13 13
14 2
15 13
16 DEST
17 After B is destructed
18 1
19 13
20 DEST
21 RELEASE

```

3.4. Memory primitives

Check out the following example:

Listing 3.7: a

```
1 struct rect<T>{
2     let a = T;
3     let b = T;
4 }
5
6 fn main() -> i32{
7     let a = in<i32>();
8     let b = in<i32>();
9
10    type irect = rect<i32>;
11
12    let rect_heap = heap_object<irect>(a,b);
13
14    out(rect_heap!.a);
15    out(rect_heap!.b);
16
17    heap_free(rect_heap);
18    out(rect_heap!.a);
19    out(rect_heap!.b);
20 }
```

3.5. Memory allocation

Check out the following example:

Listing 3.8: a

```
1 fn printpos(ptr, pos){
2     out(ptr[pos]!);
3 }
4
5 fn setpos(ptr, pos, val){
6     ptr[pos]! = val;
7 }
```

```

8
9  fn main() -> i32{
10     let A = heap_alloc<i32>(10);
11     setpos(A, 3, 123);
12     setpos(A, 6, 456);
13     printpos(A, 3);
14     printpos(A, 6);
15     printpos(A, 5);
16     heap_free(A);
17
18     // B will sometimes be allocated exactly where A was, and overwrite it
19     // of course, this is a memory error, but it will probably continue to work
20
21     let B = heap_alloc<i32>(10);
22     setpos(B, 3, 0);
23     setpos(B, 6, 0);
24     printpos(A, 3);
25     printpos(A, 6);
26     printpos(A, 5);
27
28     return 0;
29 }

```

3.6. Object oriented example

Check out the following example:

Listing 3.9: a

```

1  fn outnl(i){
2     out(i);
3     out('\n');
4  }
5
6  fn len(cptr) -> i32{
7     type _ = enable_if<cptr == @char>;
8     let cl = 0;
9     while (cptr[cl] != '\0'){
10         cl=cl+1;

```



```

11     }
12     return c1;
13 }
14
15 //
16
17 struct string{
18     let arr = @char;
19     let len = i32;
20 }
21
22 fn __init__(sptr, cptr){
23     type _ = enable_if<sptr == @string>;
24     type _ = enable_if<cptr == @char>;
25
26     let l = len(cptr);
27
28     sptr!.len = l;
29     sptr!.arr = heap_alloc<char>(l);
30     while (l>!0){
31         sptr!.arr[l-1]! = cptr[l-1]!;
32         l=l-1;
33     }
34 }
35
36 fn __copy__(sptr1, sptr2){
37     type _ = enable_if<sptr1 == @string>;
38     type _ = enable_if<sptr2 == @string>;
39
40     sptr1!.len = sptr2!.len;
41
42     let l = sptr1!.len;
43
44     sptr1!.arr = heap_alloc<char>(l);
45     while (l>!0){
46         sptr1!.arr[l-1]! = sptr2!.arr[l-1]!;
47         l=l-1;
48     }

```

```

49 }
50
51 fn __dest__( sptr ){
52     type _ = enable_if<sptr == @string>;
53
54     heap_free( sptr !. arr );
55 }
56
57 //
58
59 fn get( sptr , i ) -> char{
60     type _ = enable_if<sptr == @string>;
61     type _ = enable_if<i == i32>;
62
63     return sptr !. arr [ i ]!;
64 }
65
66 fn set( sptr , i , c ){
67     type _ = enable_if<sptr == @string>;
68     type _ = enable_if<i == i32>;
69     type _ = enable_if<c == char>;
70
71     sptr !. arr [ i ]! = c;
72 }
73
74 fn out( s ){
75     type _ = enable_if<s == string>;
76
77     for( let i=0; i <! s.len; i=i+1; ){
78         out( s . arr [ i ]! );
79     }
80 }
81
82 fn main() -> i32 {
83     let a = object<string>("Hello world!");
84     let b = a;
85
86     outnl(a);

```

```
87     outnl(b);
88     outnl(get(@a,1));
89     set(@a, 1, 'E');
90     outnl(a);
91     outnl(b);
92
93     return 0;
94 }
```

4. Future improvements

4.1. References

4.2. Syntactic improvements

4.3. Multiple source file support

4.4. Portability

5. Conclusion

Zaključak.

BIBLIOGRAPHY

- [Beazley a] BEAZLEY, David: *PLY (Python Lex-Yacc)*. – URL <http://www.dabeaz.com/ply/>
- [Beazley b] BEAZLEY, David: *PLY (Python Lex-Yacc) documentation*. – URL <https://ply.readthedocs.io/en/latest/index.html>
- [Fred L. Drake Jr.] FRED L. DRAKE JR., others: *Python3 documentation*. – URL <https://docs.python.org/3/glossary.html#term-duck-typing>
- [Lattner und Adve 2004] LATTNER, Chris ; ADVE, Vikram: *LLVM: A compilation framework for lifelong program analysis transformation*. 2004

Design of a strongly-typed programming language

Sažetak

Sažetak na hrvatskom jeziku.

Ključne riječi: Ključne riječi, odvojene zarezima.

Title

Abstract

Abstract.

Keywords: Keywords.